

DEC-08-LBAA-D

IDENTIFICATION

Product Code:	DEC-08-LBAA-D
Product Name:	Binary Loader
Date Created:	May 10, 1967
Maintainer:	Software Services Group

)

)

)

## 1. ABSTRACT

The Binary Loader is a short routine for reading and storing information contained in binary-coded tapes, using the ASR 33 Perforated-Tape Reader or the Type 750 High-Speed Perforated Tape Reader.

The Binary Loader accepts tapes prepared by the use of PAL (Program Assembly Language) or MACRO-8. Diagnostic messages may be included on tapes produced when using either PAL or MACRO. The Binary Loader will ignore all diagnostic messages.

## 2. PRELIMINARY REQUIREMENTS

### Storage

This program occupies 94 (decimal) core locations.

### Equipment

The Binary Loader may be used with a system consisting of the PDP-8 and associated Teletype ASR 33 only. On the other hand, the same program operates with systems including the 750 High-Speed Tape Reader and/or the Memory Extension Control Type 183. This loader is compatible with the 552 DECtape Library System and the TC01 DECtape Library System.

## 3. LOADING OR CALLING PROCEDURE

The Binary Loader is brought into memory by the RIM or Read-In-Mode Loader. This requires that the Binary Loader tape itself be in RIM format. See Read-In-Mode Loader Manual for a thorough discussion of the RIM Loader and RIM format.

NOTE: 183 Memory Extension users; refer to Special Requirements section.

Proceed as follows:

- a. Place the Binary Loader tape in the ASR 33 reader.
- b. Make sure that the ASR 33 is on-line.
- c. Place the starting address of the RIM Loader (7756) in the SWITCH REGISTER.
- d. Press the LOAD ADDRESS key.
- e. Press the START key.
- f. Move the READER CONTROL switch to the START position.

### Switch Setting

NOTE: 183 Memory Extension users see "Special Requirements" section.

#### 4. USING THE PROGRAM OR ROUTINE

a. Place the tape to be loaded (which must be in binary format) in either the ASR 33 Tape Reader or the Type 750 High-Speed Reader. When using the ASR 33, make sure the reader is on-line. When using the 750, make sure the reader is on.

b. Place the starting address of the Binary Loader (7777) in the SWITCH REGISTER.

c. Press LOAD ADDRESS key.

When using the 750, change the SWITCH REGISTER to 3777 (bit 0 = 0). Omit this step if using the ASR 33.

d. Press console START key.

When using the ASR 33, move the READER CONTROL switch to START.

##### Errors

When PAL is used to produce a binary tape, a checksum is automatically placed at the end of the binary tape. The checksum is the sum of all data on the tape including the origin word.

To be more specific, it is the sum of all data contained on tape that will enter the accumulator (AC) in bit positions 4 through 11 from, for example, the ASR 33 Reader buffer. Note that the sum is accumulated character by character and not word by word. Overflow (a carry out of the most-significant bit position of the AC) is ignored both when calculating a checksum (which is done by PAL) and when the Binary Loader accumulates a checksum while loading a tape.

If the checksum accumulated while using the Binary Loader does not agree with the last two characters on the tape (i.e., the checksum on the tape calculated and placed there by PAL), an error has occurred.

When the computer halts, the display lights will be static, the memory buffer (MB) will contain 7402, and the contents of the AC will be unequal to zero if a checksum error has occurred.

Restart the computer after the tape has been repositioned by pressing the CONTINUE key.

#### 5. DETAILS OF OPERATION AND STORAGE

This program furnishes the basic means by which the contents of binary-coded tapes are loaded into core.

The heart of the program is a short subroutine (tagged BEGG) which operates in outline as follows:

The incoming character is tested to see if it is a "rubout" (all eight tape channels punched).

If this is the case, all subsequent information coming from the reader is ignored until another rubout is detected.

This is the mechanism by which PAL diagnostic messages are detected. They are preceded and followed by a single rubout character. Within a diagnostic message, in contrast to the rules concerning the balance of the binary tape, any character is valid except, of course, a single rubout character itself which would prematurely conclude the diagnostic message. Note that two consecutive rubouts within a diagnostic message would, in effect, be ignored.

Next the character is tested to see if it is leader or field settings.

These tests are listed in the order in which they are performed. If none of the actions indicated have occurred upon exit from the BEGG subroutine, the character is part of the origin address, contains part of a data word, or is a part of the checksum, and the appropriate course is followed by the main routine.

## 6. SPECIAL REQUIREMENTS OR FORMATS

### 6.1 Format

#### 6.1.1 External Format - Tapes to be read by this program must be in binary-coded format.

Leader of about 1 foot of leader-trailer codes (any code with channel 8 punched; preferably code 200).

Two characters representing the address (origin) into which the first command on the next portion of the tape will be placed. Successive commands are placed in memory at addresses:

origin+1, origin+2, .....origin+n.

The initial character of the origin has no punch in channel 8, while channel 7 is punched. The second character designating the origin has no punches in either channel 8 or 7.

A concluding 2-character group representing the checksum with no punches present in channels 8 or 7.

Trailer similar to leader.

Reference to Program Listing, will indicate that after the BEGG subroutine tests to see if the character just read was leader/trailer, a test is made to determine whether the character is a "field setting." This is a reference to the fact that PAL produces tapes on which characters of the form

11 XXX 000

indicate the memory field into which the following data is to be loaded. If for example XXX were 101, all data following the field designator should be loaded into memory field five.

### 6.1.2 Example of Binary Loader Format

Tape Channel 87 654 S 321	Channels 8 and 7 Indicate	Program Proper	Notes
10 000 . 000	Leader	No	
01 000 . 010 00 000 . 000	Origin	No	In octal the origin 0200. Loading will start at 0200.
00 111 . 010 00 000 . 000	Contents of 200	Yes	The command 7200 or CLA.
00 011 . 010 00 111 . 110	Contents of 201	Yes	The command 3276 or DCA Z 076.
00 111 . 100 00 000 . 010	Contents of 202	Yes	The command 7402 or HLT.
00 000 . 100 00 010 . 010	Checksum	No	The program determines that these two characters are the checksum since trailer follows.
10 000 . 000	Trailer	No	

The octal checksum in this example is 0422. Note that this is the following sum:

102	Origin
000	
072	First word
000	
032	Second word
076	
074	Third word
<u>002</u>	
422	

### 6.2 Memory Extension Usage

6.2.1 Loading - It is recommended that the Binary Loader exist in field 0. This will insure a permanent program lining around location 7754 and 7755 which are used for TC01 DECTape. The loader will of course exist in any field, though caution must be taken not to use location 7754 and 7755 in field 0. This applies only to DECTape users. Also, when the proper field is chosen it should be noted that the RIM Loader must already be in that field.

Binary Loader Loading Procedure For Extended Memory Users

- a. Place the Binary Loader tape in the reader.
- b. Place the proper FIELD in the INSTRUCTION FIELD REGISTER when putting the starting address of the RIM Loader (7756) in the SWITCH REGISTER.
- c. Press the LOAD ADDRESS key.
- d. Press the START key.
- e. Start the reader. (ASR 33 - press READER CONTROL to start, 750 High-Speed Reader - should already be ready to start).

Operation and Usage For Extended Memory Users

- a. Place the tape to be loaded (tape must be in binary format) in the reader.  
When using the ASR 33, make sure reader is on-line. When using the 750, make sure reader is on and tape is positioned with leader/trailer over read head.
- b. In the DATA FIELD REGISTER place the field in which the program is to be loaded. In the INSTRUCTION FIELD REGISTER place the field that the binary loader is in. Place starting address of the Binary Loader (7777) in the SWITCH REGISTER.
- c. Press LOAD ADDRESS key.  
When using the 750, change the SWITCH REGISTER to 3777 (bit 0 = 0). Omit this step if using the ASR 33.
- d. Press console START key.

6.2.2 Errors - See Program Usage Section (Errors)

6.2.3 Starting of Program - After program has been successfully loaded, place starting address of program in SWITCH REGISTER. Place the field where program exists in the FIELD INSTRUCTION REGISTER.

Press LOAD ADDRESS key.

Press console START key.


7. REFERENCED MANUALS

	<u>Old Number</u>	<u>New Code</u>
RIM Loader	Digital-8-1-U	DEC-08-LRAA-D
PAL III	Digital-8-3-S	DEC-08-ASAA-D
MACRO-8	Digital-8-8-S	DEC-08-CMAA-D

8. FLOW CHARTS


Loading Binary (BIN) Loader

DEC Library Tape No: Digital-8-2-U





## Using Binary Loader



9. LISTING

/BINARY AND DECTAPE LOADERS FOR  
/555 CONTROL

*7612			
7612	0000	SWITCH,	0
7613	0000	MEMTEM,	0
7614	0000	CHAR,	0
7615	0000	CHKSUM,	0
7616	0000	ORIGIN,	0
*7626			
/EXTRACT ERRORS, FIELD, L/T			
7626	0000	BEGG,	0
7627	3212	DCA SWITCH	/SET SWITCH
7630	4260	JMS READ	/GET A CHARACTER
7631	1300	TAD M376	/TEST FOR 377
7632	7750	SPA SNA CLA	
7633	5237	JMP .+4	/NO
7634	2212	ISZ SWITCH	/YES: COMPLEMENT SWITCH
7635	7040	CMA	
7636	5227	JMP BEGG+1	
7637	1212	TAD SWITCH	/NOT 377
7640	7640	SZA CLA	/IS SWITCH SET?
7641	5230	JMP BEGG+2	/YES; IGNORE
7642	1214	TAD CHAR	/NO; TEST FOR CODE
7643	0274	AND MASK	/TYPES
7644	1341	TAD M200	
7645	7510	SPA	
7646	2226	ISZ BEGG	/DATA OR ORIGIN
7647	7750	SPA SNA CLA	
7650	5626	JMP I BEGG	/DATA, ORIGIN, or L/T
7651	1214	TAD CHAR	/FIELD SETTING
7652	0256	AND FMASK	
7653	1257	TAD CHANGE	
7654	3213	DCA MEMTEM	
7655	5230	JMP BEGG+2	/CONTINUE INPUT
7656	0070	FMASK,	70
7657	6201	CHANGE,	CDF
7660	0000	READ,	0
7661	0000		0
7662	6031	LOR,	KSF
7663	5262		JMP .-1
7664	6036		KRB
7665	3214		DCA CHAR
7666	1214		TAD CHAR
7667	5660		JMP I READ
7670	6011	HIR,	RSF
7671	5270		JMP .-1
7672	6016		RRB RFC

7673	5265		JMP LOR + 3	
7674	0300	MASK,	300	
		/TRAILER CODE SEEN		
7675	4343	BEND,	JMS ASSEMB	
7676	7041		CIA	
7677	1215		TAD CHKSUM	
7700	7402	M376,	HLT	
7701	6032	BEGIN,	KCC	
7702	6014		RFC	
7703	6214		RDF	
7704	1257		TAD CHANGE	
7705	3213		DCA MEMTEM	/SAVE FIELD INSTRUCTION
7706	7604		CLA OSR	
7707	7700		SMA CLA	
7710	1353		TAD HIRI	
7711	1352		TAD LORI	
7712	3261		DCA READ + 1	
7713	4226		JMS BEGG	
7714	5313		JMP .-1	/IGNORE LEADER
7715	3215	GO,	DCA CHKSUM	
7716	1213		TAD MEMTEM	
7717	3336		DCA MEMFLD	
7720	1214		TAD CHAR	
7721	3376		DCA WORD1	
7722	4260		JMS READ	
7723	3355		DCA WORD2	
7724	4226		JMS BEGG	/LOOK AHEAD
7725	5275		JMP BEND	/TRAILER, END
7726	4343		JMS ASSEMB	
7727	7420		SNL	
7730	5336		JMP MEMFLD	
7731	3216		DCA ORIGIN	
7732	1376	CHEX,	TAD WORD1	
7733	1355		TAD WORD2	
7734	1215		TAD CHKSUM	
7735	5315		JMP GO	
7736	0000	MEMFLD,	0	
7737	3616		DCA I ORIGIN	
7740	2216		ISZ ORIGIN	
7741	7600	M200,	7600	
7742	5332		JMP CHEX	
7743	0000	ASSEMB,	0	
7744	1376		TAD WORD1	
7745	7106		CLL RTL	
7746	7006		RTL	
7747	7006		RTL	
7750	1355		TAD WORD2	
7751	5743		JMP I ASSEMB	
7752	5262	LORI,	JMP LOR	

7753	0006	HIRI,	HIR-LOR
7754	0000		0
		WORD1=7776	
7755	0000	WORD2,	0
		*7777	
7777	5301		JMP BEGIN

ASSEMB	7743
BEGG	7626
BEGIN	7701
BEND	7675
CHANGE	7657
CHAR	7614
CHEX	7732
CHKSUM	7615
FMASK	7656
GO	7715
HIR	7670
HIRI	7753
LOR	7662
LORI	7752
MASK	7674
MEMFLD	7736
MEMTEM	7613
M200	7741
M376	7700
ORIGIN	7616
READ	7660
SWITCH	7612
WORD1	7776
WORD2	7755

NOTE: A vertical bar present in the listing before an instruction indicates a revision in the program.